Let us consider the example of a two reflector antenna. See H. Mounier, *Propriétés structurelles des systèmes linéaires à retards: aspects théoriques et pratiques*, PhD Thesis, University of Orsay, France, 1995.

```
> with(Ore_algebra):
> with(OreModules):
```

After loading the required Maple packages, the first step is to define the Ore algebra.

Let us define an Ore algebra with a differential operator Dt w.r.t. time t and a time-delay operator δ. Note also that the constants $K1$, $K2$, Te, Kp, Kc have to be declared in the definition of the Ore algebra.

```
> Alg := DefineOreAlgebra(diff=[Dt,t], dual_shift=[delta,s], polynom=[t,s],
> comm=[K1,K2,Te,Kp,Kc], shift_action=[delta,t]):
```

Enter the matrix R of the linear differential time-delay system:

```
> R := evalm([[Dt, -K1, 0, 0, 0, 0, 0, 0, 0],
> [0, Dt+K2/Te, 0, 0, 0, 0, -Kp/Te*delta, -Kc/Te*delta, -Kc/Te*delta],
> [0, 0, Dt, -K1, 0, 0, 0, 0, 0],
> [0, 0, 0, Dt+K2/Te, 0, 0, -Kc/Te*delta, -Kp/Te*delta, -Kc/Te*delta],
> [0, 0, 0, 0, Dt, -K1, 0, 0, 0],
> [0, 0, 0, 0, 0, Dt+K2/Te, -Kc/Te*delta, -Kc/Te*delta, -Kp/Te*delta]]);
```

$$R := \begin{bmatrix} Dt & -K1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & Dt+\frac{K2}{Te} & 0 & 0 & 0 & 0 & -\frac{Kp\,\delta}{Te} & -\frac{Kc\,\delta}{Te} & -\frac{Kc\,\delta}{Te} \\ 0 & 0 & Dt & -K1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & Dt+\frac{K2}{Te} & 0 & 0 & -\frac{Kc\,\delta}{Te} & -\frac{Kp\,\delta}{Te} & -\frac{Kc\,\delta}{Te} \\ 0 & 0 & 0 & 0 & Dt & -K1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & Dt+\frac{K2}{Te} & -\frac{Kc\,\delta}{Te} & -\frac{Kc\,\delta}{Te} & -\frac{Kp\,\delta}{Te} \end{bmatrix}$$

Then, we use an involution θ of Alg in order to obtain $\theta(\ R)$:

```
> R_adj := Involution(R, Alg):
```

By means of the next command, we compute the torsion-free part (if *Ext1*[1] is not the identity matrix, then the torsion part is given by *Ext1*[2]) and a parametrization of the torsion-free part in *Ext1*[3]. Equivalently, we check whether or not the two reflector antenna is controllable:

```
> st := time(): Ext1 := Exti(R_adj, Alg, 1): time() - st;
```

$$0.920$$

```
> Ext1[1];
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

We conclude that the first extension module ext^1 with values in Alg of the Alg-module N associated with *R_adj* is the zero module. Hence, the module defined by R is torsion-free. Equivalently, R is parametrizable and *Ext1*[3] gives a parametrization of R which involves three free parameters:

```
> Ext1[3];
```

$$\begin{bmatrix} K1\,\delta\,Kc & K1\,\delta\,Kc & Kp\,K1\,\delta \\ Dt\,\delta\,Kc & Dt\,\delta\,Kc & Kp\,\delta\,Dt \\ K1\,\delta\,Kc & Kp\,K1\,\delta & K1\,\delta\,Kc \\ Dt\,\delta\,Kc & Kp\,\delta\,Dt & Dt\,\delta\,Kc \\ Kp\,K1\,\delta & K1\,\delta\,Kc & K1\,\delta\,Kc \\ Kp\,\delta\,Dt & Dt\,\delta\,Kc & Dt\,\delta\,Kc \\ 0 & 0 & Dt^2\,Te + Dt\,K2 \\ 0 & Dt^2\,Te + Dt\,K2 & 0 \\ Dt^2\,Te + Dt\,K2 & 0 & 0 \end{bmatrix}$$

The same parametrization can be obtained by using *Parametrization*. The result involves three free functions ξ_1, ξ_2, ξ_3:

```
> Parametrization(R, Alg);
```

$$\begin{bmatrix} K1\,Kc\,\xi_1(t-1) + K1\,Kc\,\xi_2(t-1) + Kp\,K1\,\xi_3(t-1) \\ Kc\,\mathrm{D}(\xi_1)(t-1) + Kc\,\mathrm{D}(\xi_2)(t-1) + Kp\,\mathrm{D}(\xi_3)(t-1) \\ K1\,Kc\,\xi_1(t-1) + Kp\,K1\,\xi_2(t-1) + K1\,Kc\,\xi_3(t-1) \\ Kc\,\mathrm{D}(\xi_1)(t-1) + Kp\,\mathrm{D}(\xi_2)(t-1) + Kc\,\mathrm{D}(\xi_3)(t-1) \\ Kp\,K1\,\xi_1(t-1) + K1\,Kc\,\xi_2(t-1) + K1\,Kc\,\xi_3(t-1) \\ Kp\,\mathrm{D}(\xi_1)(t-1) + Kc\,\mathrm{D}(\xi_2)(t-1) + Kc\,\mathrm{D}(\xi_3)(t-1) \\ Te\,(\mathrm{D}^{(2)})(\xi_3)(t) + K2\,\mathrm{D}(\xi_3)(t) \\ Te\,(\mathrm{D}^{(2)})(\xi_2)(t) + K2\,\mathrm{D}(\xi_2)(t) \\ Te\,(\mathrm{D}^{(2)})(\xi_1)(t) + K2\,\mathrm{D}(\xi_1)(t) \end{bmatrix}$$

The two reflector antenna is not a flat system because the second extension module extˆ2 with values in *Alg* of the *Alg*-module N is different from zero, as shown next:

```
> st := time(): Ext2 := Exti(R_adj, Alg, 2): time() - st;
```

$$0.750$$

```
> Ext2[1];
```

$$\begin{bmatrix} \delta & 0 & 0 \\ Dt^2\,Te + Dt\,K2 & 0 & 0 \\ 0 & \delta & 0 \\ 0 & Dt^2\,Te + Dt\,K2 & 0 \\ 0 & 0 & \delta \\ 0 & 0 & Dt^2\,Te + Dt\,K2 \end{bmatrix}$$

Since the *torsion-free degree* $\mathrm{i}(M)$ of M is equal to 1 (i.e., M is a torsion-free but not a free *Alg*-module), we can find a polynomial π in the variable δ such that the system is π-free:

```
> PiPolynomial(R, Alg, [delta]);
```

$$[\delta]$$

By definition of the π-polynomial (Mounier, 1995), this means that if we introduce the time-advance operator in the system of the two reflector antenna, then it becomes a flat system. Hence, the module associated with this system is a free module (over the Ore algebra which is obtained by adjoining the advance operator δ^{-1} to *Alg*). We shall find a basis for this module below.

Let us remark that the fact that the two reflector antenna is not a flat system (without advance operator) is coherent with the fact that the full row-rank matrix R does not admit a right-inverse. We remember that a full row-rank matrix R admits a right-inverse if and only if the module which is associated with it is projective. By the theorem of Quillen-Suslin, for modules over commutative polynomial rings, projectiveness is the same as freeness. This remark applies to our situation as we have:

```
> SyzygyModule(R, Alg); RightInverse(R, Alg);
```

$$\mathrm{INJ}(6)$$

$$[]$$

The fact that the system is not flat is also coherent with the fact that its parametrization *Ext1*[3] does not admit a left-inverse: a linear system is flat if and only if it is parametrizable and one of its parametrization admits a left-inverse.

```
> LeftInverse(Ext1[3], Alg);
```

$$[]$$

We finish by computing a basis of the free module M_2 which is associated to the system of the two reflector antenna, when introducing the time-advance operator in the Ore algebra *Alg*. In the terminology of control, such a basis is called a *flat output*. We apply *LocalLeftInverse* to the parametrization *Ext1*[3] of the system and allow the algorithm to invert δ:

```
> S := LocalLeftInverse(Ext1[3], [delta], Alg);
```

$$S := \begin{bmatrix} -\frac{Kc}{\delta\, K1\, \%1} & 0 & -\frac{Kc}{\delta\, K1\, \%1} & 0 & \frac{Kp+Kc}{\delta\, K1\, \%1} & 0 & 0 & 0 & 0 \\ -\frac{Kc}{\delta\, K1\, \%1} & 0 & \frac{Kp+Kc}{\delta\, K1\, \%1} & 0 & -\frac{Kc}{\delta\, K1\, \%1} & 0 & 0 & 0 & 0 \\ \frac{Kp+Kc}{\delta\, K1\, \%1} & 0 & -\frac{Kc}{\delta\, K1\, \%1} & 0 & -\frac{Kc}{\delta\, K1\, \%1} & 0 & 0 & 0 & 0 \end{bmatrix}$$
$$\%1 := Kp^2 - 2\,Kc^2 + Kp\,Kc$$

By construction, the matrix S is a left-inverse of *Ext1*[3]:

```
> simplify(evalm(S &* Ext1[3]));
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Therefore, $(z_1 : z_2 : z_3)^T = S\,(x_1 : \ldots : x_6 : u_1 : u_2 : u_3)^T$ is a basis of the $Alg[\delta^{-1}]$-module M_2 associated with R, and thus, a flat output of the two reflector antenna, when we introduce the time-advance operator. More precisely, a flat output $(z_1 : z_2 : z_3)^T$ of the system is defined by:

```
> evalm([seq([z[i](t)],i=1..3)])=ApplyMatrix(S,[seq(x[i](t),i=1..6),
> seq(u[i](t),i=1..3)],Alg);
```

$$\begin{bmatrix} z_1(t) \\ z_2(t) \\ z_3(t) \end{bmatrix} = \begin{bmatrix} -\frac{Kc\,x_1(t+1)}{K1\,\%1} - \frac{Kc\,x_3(t+1)}{K1\,\%1} + \frac{(Kc+Kp)\,x_5(t+1)}{K1\,\%1} \\ -\frac{Kc\,x_1(t+1)}{K1\,\%1} + \frac{(Kc+Kp)\,x_3(t+1)}{K1\,\%1} - \frac{Kc\,x_5(t+1)}{K1\,\%1} \\ \frac{(Kc+Kp)\,x_1(t+1)}{K1\,\%1} - \frac{Kc\,x_3(t+1)}{K1\,\%1} - \frac{Kc\,x_5(t+1)}{K1\,\%1} \end{bmatrix}$$
$$\%1 := Kp\,Kc - 2\,Kc^2 + Kp^2$$

Finally, if we substitute $(z_1 : z_2 : z_3)^T$ into the parametrization *Ext1*[3] of the system, we obtain $(x_1 : \ldots : x_6 : u_1 : u_2 : u_3)^T = T(x_1 : \ldots : x_6 : u_1 : u_2 : u_3)^T$, where the matrix T is defined by:

```
> T := simplify(evalm(Ext1[3] &* S));
```

$$T := \left[\begin{array}{c} 1\,,0\,,0\,,0\,,0\,,0\,,0\,,0\,,0 \\ \dfrac{Dt}{K1}\,,0\,,0\,,0\,,0\,,0\,,0\,,0\,,0 \\ 0\,,0\,,1\,,0\,,0\,,0\,,0\,,0\,,0 \\ 0\,,0\,,\dfrac{Dt}{K1}\,,0\,,0\,,0\,,0\,,0\,,0 \\ 0\,,0\,,0\,,0\,,1\,,0\,,0\,,0\,,0 \\ 0\,,0\,,0\,,0\,,\dfrac{Dt}{K1}\,,0\,,0\,,0\,,0 \\ \dfrac{Dt\,(Dt\;Te+K2)\,(Kp+Kc)}{\delta\,K1\,\%1}\,,0\,,\%2\,,0\,,\%2\,,0\,,0\,,0\,,0 \\ \%2\,,0\,,\dfrac{Dt\,(Dt\;Te+K2)\,(Kp+Kc)}{\delta\,K1\,\%1}\,,0\,,\%2\,,0\,,0\,,0\,,0 \\ \%2\,,0\,,\%2\,,0\,,\dfrac{Dt\,(Dt\;Te+K2)\,(Kp+Kc)}{\delta\,K1\,\%1}\,,0\,,0\,,0\,,0 \end{array}\right]$$

$$\%1 := Kp^2 - 2\,Kc^2 + Kp\,Kc$$

$$\%2 := -\frac{Dt\,(Dt\;Te+K2)\,Kc}{\delta\,K1\,\%1}$$

We notice that $(x_2 : x_4 : x_6 : u_1 : u_2 : u_3)^T$ is expressed in terms of x_1, x_3 and x_5 only. Thus, $(x_1 : x_3 : x_5)$ is also a basis of the $Alg[\delta^{-1}]$-module M_2 (compare with (Mounier, 1995)). More precisely, we have:

```
> evalm([seq([x[i](t)=ApplyMatrix(T, [seq(x[j](t), j=1..6),seq(u[j](t), j=1..3)],
> Alg)[i,1]], i=1..6)]);
```

$$\left[\begin{array}{c} x_1(t) = x_1(t) \\ x_2(t) = \dfrac{\mathrm{D}(x_1)(t)}{K1} \\ x_3(t) = x_3(t) \\ x_4(t) = \dfrac{\mathrm{D}(x_3)(t)}{K1} \\ x_5(t) = x_5(t) \\ x_6(t) = \dfrac{\mathrm{D}(x_5)(t)}{K1} \end{array}\right]$$

```
> evalm([seq([u[i](t)=ApplyMatrix(T, [seq(x[j](t), j=1..6),seq(u[j](t), j=1..3)],
> Alg)[6+i,1]], i=1..3)]);
```

$$\left[u_1(t) = \frac{\mathit{K2}\,(Kc + Kp)\,\mathrm{D}(x_1)(t+1)}{\mathit{K1}\,\%1} + \frac{\mathit{Te}\,(Kc + Kp)\,(\mathrm{D}^{(2)})(x_1)(t+1)}{\mathit{K1}\,\%1}\right.$$
$$- \frac{\mathit{K2}\,Kc\,\mathrm{D}(x_3)(t+1)}{\mathit{K1}\,\%1} - \frac{\mathit{Te}\,Kc\,(\mathrm{D}^{(2)})(x_3)(t+1)}{\mathit{K1}\,\%1} - \frac{\mathit{K2}\,Kc\,\mathrm{D}(x_5)(t+1)}{\mathit{K1}\,\%1}$$
$$\left.- \frac{\mathit{Te}\,Kc\,(\mathrm{D}^{(2)})(x_5)(t+1)}{\mathit{K1}\,\%1}\right]$$
$$\left[u_2(t) = -\frac{\mathit{K2}\,Kc\,\mathrm{D}(x_1)(t+1)}{\mathit{K1}\,\%1} - \frac{\mathit{Te}\,Kc\,(\mathrm{D}^{(2)})(x_1)(t+1)}{\mathit{K1}\,\%1} + \frac{\mathit{K2}\,(Kc + Kp)\,\mathrm{D}(x_3)(t+1)}{\mathit{K1}\,\%1}\right.$$
$$\left.+ \frac{\mathit{Te}\,(Kc + Kp)\,(\mathrm{D}^{(2)})(x_3)(t+1)}{\mathit{K1}\,\%1} - \frac{\mathit{K2}\,Kc\,\mathrm{D}(x_5)(t+1)}{\mathit{K1}\,\%1} - \frac{\mathit{Te}\,Kc\,(\mathrm{D}^{(2)})(x_5)(t+1)}{\mathit{K1}\,\%1}\right]$$
$$\left[u_3(t) = -\frac{\mathit{K2}\,Kc\,\mathrm{D}(x_1)(t+1)}{\mathit{K1}\,\%1} - \frac{\mathit{Te}\,Kc\,(\mathrm{D}^{(2)})(x_1)(t+1)}{\mathit{K1}\,\%1} - \frac{\mathit{K2}\,Kc\,\mathrm{D}(x_3)(t+1)}{\mathit{K1}\,\%1}\right.$$
$$- \frac{\mathit{Te}\,Kc\,(\mathrm{D}^{(2)})(x_3)(t+1)}{\mathit{K1}\,\%1} + \frac{\mathit{K2}\,(Kc + Kp)\,\mathrm{D}(x_5)(t+1)}{\mathit{K1}\,\%1}$$
$$\left.+ \frac{\mathit{Te}\,(Kc + Kp)\,(\mathrm{D}^{(2)})(x_5)(t+1)}{\mathit{K1}\,\%1}\right]$$
$$\%1 := Kp\,Kc - 2\,Kc^2 + Kp^2$$